

TRIAL STATISTICAL ANALYSIS PLAN

PERI-INTERVENTIONAL OUTCOME STUDY IN THE ELDERLY (POSE):
EUROPEAN, MULTI-CENTRE, PROSPECTIVE OBSERVATIONAL
COHORT STUDY

VERSION 1.1, 20.01.2020

Clinical Trials.gov	NCT03152734
Chief Coordinating Investigator	Prof. Dr. Mark Coburn, MD Managing Senior Physician Department of Anaesthesia Medical Faculty, RWTH Aachen University, Germany
Biostatistician	Prof. Dr. Ralf-Dieter Hilgers, PhD Head of the Department of Medical Statistics Medical Faculty, RWTH Aachen University, Germany Marcia Viviane Rückbeil, M. Sc. Department of Medical Statistics Medical Faculty, RWTH Aachen University, Germany
Central Organisation/ Project Management	Dr. Ana Kowark (née Stevanovic), MD Department of Anaesthesia Medical Faculty, RWTH Aachen University, Germany

SUMMARY

LIST OF ABBREVIATIONS.....	4
1. INTRODUCTION	5
2. SOFTWARE UTILIZED	5
2.1 OPERATING SYSTEM	5
2.2 DATABASE.....	5
2.3 STATISTICAL ANALYSIS.....	5
3. STUDY DESIGN.....	5
4. STUDY OBJECTIVES	5
5. TARGET VARIABLES	5
5.1 PRIMARY ENDPOINT	5
5.2 SECONDARY ENDPOINTS	6
5.2.1 <i>Continuous secondary endpoints</i>	7
5.2.2 <i>Dichotomous secondary endpoints</i>	7
5.2.3 <i>Nominal secondary endpoints</i>	7
5.2.4 <i>Ordinal secondary endpoints</i>	7
5.3 SENSITIVITY ANALYSES	7
6. DATA HANDLING	8
6.1 DATE AND TIME VARIABLES.....	8
6.2 PRECISION OF ESTIMATES	8
6.3 TRANSFORMATION OF HEAVILY SKEWED DATA.....	8
6.4 DEFINITION OF THE PRIMARY ENDPOINT	8
6.5 DEFINITION OF SECONDARY ENDPOINTS	9
6.6 CATEGORIZATION OF INDEPENDENT VARIABLES.....	10
6.6.1 <i>Age categories</i>	10
6.6.2 <i>Intervention technique categories</i>	10
6.6.3 <i>Type of anaesthesia</i>	11
6.7 DEFINITION OF INDEPENDENT VARIABLES.....	11
6.7.1 <i>Frailty</i>	11
6.7.2 <i>Multimorbidity</i>	12
6.7.3 <i>Timed up and go test result</i>	13
6.8 MISSING DATA	14
7. DESCRIPTIVE STATISTICS	15
7.1 BASELINE CHARACTERISTICS	15
7.2 PRIMARY ENDPOINT	15
7.3 SECONDARY ENDPOINTS	16
7.3.1 <i>Continuous secondary endpoints</i>	16
7.3.2 <i>Dichotomous secondary endpoints</i>	16
7.3.3 <i>Nominal secondary endpoints</i>	16

7.3.4	Ordinal secondary endpoints.....	16
7.4	SENSITIVITY ANALYSES	16
8.	INFERENCE STATISTICS	17
8.1	PRIMARY ENDPOINT	17
9.	PREPARATION AND QUALITY CONTROL OF PROGRAM CODE.....	20
9.1	STORAGE OF LOG FILES.....	20
9.2	CONTROL OF PROGRAM	20
10.	CHANGES FROM STUDY PROTOCOL.....	20
11.	REFERENCES	21

LIST OF ABBREVIATIONS

ACS	American College of Surgeons
AKI	Acute kidney injury
ASA	American Society of Anesthesiologists
BSCI	Brief Screen for Cognitive Impairment
CRF	Case Report Form
ICH	International Declaration of Helsinki
ICU	Intensive care unit
LOS	Length of stay
NSQIP	National Surgical Quality Improvement Program
RWTH	Rheinisch-Westfälische Technische Hochschule
USA	United States of America
WHO	World Health Organization

1. INTRODUCTION

The Trial Statistical Analysis Plan (TSAP) is a detailed technical extension to the clinical study protocol focussing on data handling and statistical analysis, which follows the principles of the guidelines ICH E3, E6, E8, E9 and considers recommendations of the STROBE statement for cohort studies. It is related to:

Document	Version	Relation
Study Protocol	Version 2.0, 14.06.2018	Description of study design and objectives

The study protocol can be accessed through <https://pose-trial.org/study-documents/>

This TSAP was written before analysing the data. It is limited to the description of the analysis of the primary outcome.

2. SOFTWARE UTILIZED

2.1 OPERATING SYSTEM

Current operating system is Windows 10 Professional. In the event of an update to a new version of the operating system, the new system is then used.

2.2 DATABASE

OpenClinica, Version 3.14.

2.3 STATISTICAL ANALYSIS

SAS (Version 9.4, SAS Institute Inc, Cary, NC, USA) and R Version 3.5.1 are used for statistical analyses. In the event of an update to a new version of SAS or R, the new version is then used.

3. STUDY DESIGN

The POSE-trial is a European multi-centre, prospective observational cohort study.

4. STUDY OBJECTIVES

An important goal of the POSE-trial is the description of the current health care of elderly patients. Therefore, many of the collected endpoints will be presented descriptively. The primary endpoint (time until all-cause mortality with maximum follow-up of 30 days) will additionally be analysed using inferential statistics.

5. TARGET VARIABLES

A detailed description of all variables can be found in the CRFs. The CRFs can be accessed through <https://pose-trial.org/study-documents/>

5.1 PRIMARY ENDPOINT

The primary endpoint is defined as the time after intervention until all-cause mortality (death from any cause). The maximum length of follow-up is set to 30 days.

Variables regarding the primary endpoint are given in Table 1. The exact definition of the primary endpoint (variables Surv_Time_30 and Surv_Status_30) is explained in section 6.4.

Variable	Description	Unit	Data type
Surv_Time_30	Survival time until all-cause mortality or censoring	Days	Continuous
Surv_Status_30	Censoring status (1 = patient died)		Dichotomous

Table 1: Primary endpoint.

5.2 SECONDARY ENDPOINTS

On overview of all secondary endpoints is given in Table 2. A description of secondary endpoints classified by data type follows in subsections 5.2.1 to 5.2.4. The exact definition of some of the secondary endpoints is explained in section 6.5.

Variable	Description	Data type
Discharge_Destination	Discharge destination category if patient was discharged before day 30	Nominal
Hospital_Length_Stay	Hospital length of stay after intervention until follow-up	Continuous
ICU_Length_Stay	Total ICU length of stay after intervention	Continuous
Admission_ICU_FollowUp	Unplanned ICU admission at any time after intervention until day 30	Dichotomous
Admission_Geriatric_FollowUp	Admission to a unit with "geriatric support" at any time-point until day 30	Dichotomous
NSQIP_Outcome_Hospital	In hospital outcome according to the ACS NSQIP (None of the mentioned, Cardiac arrest, ...)	Nominal
NSQIP_none	NSQIP: None of the mentioned	Dichotomous
NSQIP_Cardiac_Arrest	NSQIP: Cardiac arrest	Dichotomous
NSQIP_Myocard_Infarct	NSQIP: Myocardial infarction	Dichotomous
NSQIP_Pneumonia	NSQIP: Pneumonia	Dichotomous
NSQIP_Pulm_Embol	NSQIP: Pulmonary embolism	Dichotomous
NSQIP_Unplanned_Intub	NSQIP: Unplanned intubation	Dichotomous
NSQIP_Ventilator_48h	NSQIP: Ventilator > 48h	Dichotomous
NSQIP_Return_OP	NSQIP: Return to the operating room	Dichotomous
NSQIP_Stroke	NSQIP: Stroke	Dichotomous
NSQIP_AKI	NSQIP: Acute kidney injury	Dichotomous
NSQIP_DVT	NSQIP: DVT	Dichotomous
NSQIP_Venous_Thromb	NSQIP: Venous thromboembolism/ blood clot	Dichotomous
NSQIP_Super_SSI	NSQIP: Superficial incisional SSI	Dichotomous
NSQIP_Deep_SSI	NSQIP: Deep incisional SSI	Dichotomous
NSQIP_Organ_SSI	NSQIP: Organ space SSI	Dichotomous
NSQIP_Wound_Disruption	NSQIP: Wound disruption	Dichotomous
NSQIP_Systemic_Sepsis	NSQIP: Systemic sepsis	Dichotomous
NSQIP_Urinary_Infect	NSQIP: Urinary tract infection	Dichotomous
NSQIP_Discharge_PA_Care	NSQIP: Discharge to post-acute care (other hospital, nursing /rehab facility)	Dichotomous
Functional_Status_FollowUp	Actual functional status in hospital	Ordinal
BSCI_FollowUp	Brief screen for cognitive impairment at Hospital: Correct Number of recalled words	Continuous
FollowUp_Status	Patient status on day 30 (Alive, Dead, Follow-up not performed/not available)	Nominal
Complications_After_Discharge	Any complications after hospital discharge requiring re-admission	Dichotomous
Cardiac_Complications	Cardiac complications after hospital discharge requiring re-damission	Dichotomous
Pulmonary_Complications	Pulmonary complications after hospital discharge requiring re-damission	Dichotomous

Stroke_Complications	Stroke after hospital discharge requiring re-admission	Dichotomous
AKI_Complications	Acute kidney injury requiring renal replacement therapy	Dichotomous
Functional_Status_TFU	Actual functional status in telephone interview	Ordinal
BSCI_TFU	Brief screen for cognitive impairment at telephone interview: Correct Number of recalled words	Continuous

Table 2: Secondary endpoints.

5.2.1 Continuous secondary endpoints

Continuous secondary endpoints include:

- Duration of anaesthesia during intervention
- Hospital length of stay after intervention
- Total ICU length of stay after intervention
- The cognitive status measured by the brief screen for cognitive impairment (scale from 0-3) measured at follow-up

5.2.2 Dichotomous secondary endpoints

Dichotomous secondary endpoints include:

- Admission to ICU directly after intervention
- If admission to ICU directly after intervention, was the admission unplanned
- Admission to a unit with "geriatric" support directly after intervention
- Intraoperative transfusion of packed blood cells, fresh frozen plasma or platelets
- The occurrence of unplanned admission to ICU at any time after intervention until day 30
- The occurrence of admission to a unit with "geriatric support" at any time after intervention until day 30
- In hospital outcomes according to the ACS NSQIP (such as cardiac arrest, myocardial infarction, pneumonia, pulmonary embolism) after intervention until day 30
- The occurrence of unplanned re-intubation within the hospital stay or maximum 30 days post-interventional.
- The occurrence of peri-interventional major complications after hospital discharge that led to hospital re-admission (such as serious cardiac, pulmonary complications or stroke) or to renal replacement therapy (acute kidney injury)

5.2.3 Nominal secondary endpoints

Nominal secondary endpoints include:

- Hospital discharge destination (Other hospital, rehabilitation, nursing home, home, other)

5.2.4 Ordinal secondary endpoints

Ordinal secondary endpoints include:

- The functional status (independent < partially dependent < totally dependent) measured at follow-up

5.3 SENSITIVITY ANALYSES

We will perform the following sensitivity analyses:

- We will perform a sensitivity analysis of the primary endpoint (time until all-cause mortality) with the following two-way interaction effects:
 - Age and premedication before intervention
 - Frailty and premedication
 - Anaesthesia technique and severity of intervention

The interaction effects were chosen based on clinical relevance. For the sensitivity analysis, the premedication before intervention and the anaesthesia technique are included as independent variables even if they are not selected as independent variables for the main analysis. All other independent variables are the same as in the main analysis.

- We will perform a sensitivity analysis of the primary endpoint (time until all-cause mortality), treating all missing data from covariates as missing, i.e., without using imputation techniques. The independent variables are the same as in the main analysis.

6. DATA HANDLING

Beside the follow up contact on post-interventional day 30, all data will be collected during the clinical routine and there will be no study-related interventions.

After closing of the OpenClinica database, the database will be exported in SAS format. Access to the study data can be requested from the Department of Anaesthesia (Medical Faculty, RWTH Aachen University). Requests will be handled on a case-by-case basis.

6.1 DATE AND TIME VARIABLES

Date Variables are transformed to a SAS defined numeric variable with SAS Format DATE9. Time variables are transformed to a SAS defined numeric variable with SAS Format TIME5.

6.2 PRECISION OF ESTIMATES

All summary statistics (e.g. mean, SD) will be rounded to one decimal place more than decimal places given in export from OpenClinica. Percentages will be rounded to two decimal places. Estimates and p-values will be rounded to four decimal places.

Because of possible rounding errors, the sum of percentages is not necessarily 100 %.

6.3 TRANSFORMATION OF HEAVILY SKEWED DATA

Heavily skewed continuous variables will be transformed for inferential statistic to improve model fit. The skewness of continuous variables will be investigated using histograms and boxplots by visual inspection.

6.4 DEFINITION OF THE PRIMARY ENDPOINT

The primary endpoint of time until death (all-cause mortality) is defined by a survival time and a status variable. The status variable is 1 if the patient experiences an event (dies) and 0 if the patient does not have an event (observation is censored). The survival time is the time until death or the time until last contact (censoring) with a maximum follow-up of 30 days.

Variable	Description	Unit	Data type
Date_Intervention_End	Date of the end of anaesthesia (intervention)	DDMMYY	DATE
Date_FollowUp	Date the follow-up was performed	DDMMYY	DATE
Death_Hospital	Did patient die in hospital before day 30		Dichotomous
Date_Death_Hospital	Date of death in hospital	DDMMYY	DATE

Discharged	Was patient discharged before day 30		Dichotomous
Date_Discharge	Date of discharge	DDMMYY	DATE
FollowUp_Status	Patient status on day 30 (Alive, Dead, Follow-up not performed/not available)		Nominal
Date_Death_Outside	Date of death outside	DDMMYY	DATE
Surv_Time	Time until event or censoring for maximum available follow-up	Days	Continuous
Surv_Status	Censoring status for maximum available follow-up		Dichotomous
Surv_Time_30	Time until event or censoring for maximum follow-up of 30 days	Days	Continuous
Surv_Status_30	Censoring status for maximum follow-up of 30 days		Dichotomous

Table 3: Variables for the definition of the primary endpoint.

```

/* Define primary endpoint (time until death) */
DATA poseData; SET poseData;
  Surv_Time = . ; Surv_Status = .;
  * Patient still in hospital at day 30, alive;
  if Discharged = 2 and Death_Hospital = 2 then do;
    Surv_Status = 0;
    Surv_Time = DATDIF(Date_Intervention_End, Date_FollowUp, 'ACT/ACT');
  end;
  * Patient died in hospital;
  if Death_Hospital = 1 then do;
    Surv_Status = 1;
    Surv_Time = DATDIF(Date_Intervention_End, Date_Death_Hospital, 'ACT/ACT');
  end;
  * Patient discharged before day 30, died outside;
  if Discharged = 1 and FollowUp_Status = 2 then do;
    Surv_Status = 1;
    Surv_Time = DATDIF(Date_Intervention_End, Date_Death_Outside, 'ACT/ACT');
  end;
  * Patient discharged before day 30, still alive at telephone interview;
  if Discharged = 1 and FollowUp_Status = 1 then do;
    Surv_Status = 0;
    Surv_Time = DATDIF(Date_Intervention_End, Date_FollowUp, 'ACT/ACT');
  end;
  * Patient discharged before day 30, no follow-up;
  if Discharged = 1 and FollowUp_Status = 3 then do;
    Surv_Status = 0;
    Surv_Time = DATDIF(Date_Intervention_End, Date_Discharge, 'ACT/ACT');
  end;
  * Other scenarios should not be observed;
  else do;
    put 'Invalid category for' Surv_Stat;
    put 'Invalid category for' Surv_Time;
  end;
RUN;

/* Censor deaths due to the maximum follow-up of 30 days */
DATA poseData; SET poseData;
  Surv_Time_30 = min(Surv_Time, 30);
  Surv_Status_30 = Surv_Status;
  if Surv_Time > 30 and Surv_Status = 1 then do;
    Surv_Time_30 = 30;
    Surv_Status_30 = 0;
  end;
RUN;

```

6.5 DEFINITION OF SECONDARY ENDPOINTS

The duration of anaesthesia (Duration_Anaesthesia) is defined as the difference in minutes between start and end of anaesthesia.

Variable	Description	Unit	Data type
Date_Intervention_Start	Date of the start of anaesthesia (intervention)	DDMMYY	DATE
Anaesthesia_Start	Start of anaesthesia time	hh:mm	TIME
Date_Intervention_End	Date of the end of anaesthesia (intervention)	DDMMYY	DATE
Anaesthesia_End	End of anaesthesia time	hh:mm	TIME
Duration_Anaesthesia	Duration of anaesthesia	Minutes	Continuous

Table 4: Variables for the definition of the duration of anaesthesia endpoints.

```

/* Define duration of anaesthesia */
DATA poseData; SET poseData;
    Duration_Anaesthesia = INTCK('minute', Anaesthesia_Start, Anaesthesia_End) +
    1440*(Date_Intervention_End-Date_Intervention_Start);
RUN;
    
```

6.6 CATEGORIZATION OF INDEPENDENT VARIABLES

Some continuous variables are categorized for description (e.g. Kaplan-Meier curves). Continuous variables will, however not be categorized for inference on primary or secondary outcomes and treated as continuous covariates.

Some nominal variables are categorized to create variables with fewer categories. The newly created variables with fewer categories are used for inference.

6.6.1 Age categories

For description, the age of patients is categorized in categories from 80 to 84, 85 to 89 and 90 to 120.

Variable	Description	Unit	Data type
Age	Age	years	Continuous
Age_Category	Age categorized in 3 groups		Ordinal

Table 5: Variables for the categorization of age.

```

/* Define age categories */
DATA poseData; SET poseData;
    length Age_Category $12;
    if Age >=80 and Age<85 then Age_Category = "[80,85)";
    if Age >=85 and Age<90 then Age_Category = "[85,90)";
    if Age >=90 then Age_Category = "> 90";
RUN;
    
```

6.6.2 Intervention technique categories

The intervention technique was surveyed in 26 categories. For statistical inference, these 26 categories are grouped into 9 main categories:

1. Cardiovascular and Thoracic: Cardiac; Thoracic; Vascular major; Vascular minor
2. Gynaecologic and Urologic: Gynaecologic; Urologic major; Urologic minor
3. Abdominal: Gastrointestinal major; Gastrointestinal minor; Hepatic major; Hepatic minor
4. Orthopaedic, Trauma and Plastic: Athroplasty and spine; Multiple trauma related; Orthopaedic other; Orthopaedic trauma; Plastic
5. ENT and Eyes: Ear, nose and throat (ENT); Ophthalmologic
6. Interventional: Endoscopic digestive; Interv. Cardiology; Interv. Cardiorhythmology; Interv. neuroradiology
7. Neurosurgery: Neuro
8. Other surgery

9. Transplant: Transplant; Renal transplant

Variable	Description	Unit	Data type
Interventional_Category_26	Different interventional techniques surveyed in 26 groups		Nominal
Interventional_Category_9	Different interventional techniques categorized in 9 groups		Nominal

Table 6: Variables for the categorization of interventional techniques.

```

/* Define surgical categories */
DATA poseData; SET poseData;
length Interventional_Category_9 $30;
Interventional_Category_9 = '';
if Interventional_Category_26 in ('2', '20', '24', '25') then
    Interventional_Category_9 = 'Cardiovasc and thoracic';
if Interventional_Category_26 in ('7', '22', '23') then
    Interventional_Category_9 = 'Gyn and uro';
if Interventional_Category_26 in ('5', '6', '8', '9') then
    Interventional_Category_9 = 'Abdominal';
if Interventional_Category_26 in ('1', '13', '16', '17', '18') then
    Interventional_Category_9 = 'Ortho, trauma and plastic';
if Interventional_Category_26 in ('3', '15') then Interventional_Category_9 =
    'ENT and Eyes';
if Interventional_Category_26 in ('4', '10', '11', '12') then
    Interventional_Category_9 = 'Interventional';
if Interventional_Category_26 in ('14') then Interventional_Category_9 =
    'Neurosurgery';
if Interventional_Category_26 in ('19', '21') then Interventional_Category_9 =
    'Transplant';
if Interventional_Category_26 in ('26') then Interventional_Category_9 =
    'Other';
RUN;
    
```

6.6.3 Type of anaesthesia

The type of anaesthesia is categorized in 4 groups (general anaesthesia vs. regional anaesthesia vs. sedation anaesthesia vs. combined anaesthesia).

Variable	Description	Unit	Data type
Anaesthesia_Techniques	List of anaesthesia techniques used		Nominal
Anaesthesia_Technique	Different anaesthesia techniques categorized in 4 groups of anaesthesia		Nominal

Table 7: Variables for the categorization of anaesthesia techniques.

```

/* Define anaesthesia categories */
DATA poseData; SET poseData;
Anaesthesia_Technique = "Combination";
if Anaesthesia_Techniques = "1" then Anaesthesia_Technique = "General";
if Anaesthesia_Techniques in ("2", "3", "4") then Anaesthesia_Technique =
    "Regional";
if Anaesthesia_Techniques = "5" then Anaesthesia_Technique = "Sedation";
if Anaesthesia_Techniques = "" then Anaesthesia_Technique = "";
RUN;
    
```

6.7 DEFINITION OF INDEPENDENT VARIABLES

6.7.1 Frailty

Frailty is defined similar to (Oresanya, Lyons and Finlayson 2014), which is based on the original publication by (Robinson, et al. 2009). A patient is classified as frail if at least 4 of the following 6 criteria are fulfilled:

- Mini-Cog score of ≤ 3 points
- Albumin level of ≤ 3.3 g/dl

- More than 1 fall in the last 6 months
- Haematocrit level of < 35%
- Preoperative functional status is partially dependent or totally dependent
- ≥3 comorbidities

By definition, if one criterion is missing, the criterion is interpreted as “not present”.

Variable	Description	Unit	Data type
Mini_Cog_Total	Total point from the Mini Cog test at baseline		Continuous
Albumin	Albumin level at baseline	g/dl	Continuous
History_Falls	History of falls during the last 6 months		Ordinal
Haematocrit	Haematocrit level at baseline	%	Continuous
Functional_Status_BL	Functional status within 30 days before baseline		Ordinal
Risk_Ischemic_Heart_Disease	Presence of Ischemic heart disease		Dichotomous
Risk_Cardiac_Arrhythmia	Presence of cardiac arrhythmia or heart blocks		Dichotomous
Risk_Chronic_Heart_Failure	Presence of chronic heart failure or cardiomyopathy		Dichotomous
Risk_Peri_Vascular_Disease	Presence of peripheral vascular disease		Dichotomous
Risk_Hemiplegia	Presence of hemiplegia		Dichotomous
Risk_COPD	Presence of COPD		Dichotomous
Risk_Chronic_Resp_Failure	Presence of chronic respiratory failure		Dichotomous
Risk_Alcohol_Abuse	Chronic alcohol abuse		Dichotomous
Risk_Cancer	Presence of cancer		Dichotomous
Risk_Transpl_Organs	Presence of transplanted organ(s)		Dichotomous
Risk_Dementia	Presence of dementia		Dichotomous
Risk_Cerebrovascular_Disease	Presence of cerebrovascular disease		Dichotomous
Risk_Mild_Cogn_Impair	Presence of mild cognitive impairment		Dichotomous
Risk_Chron_Renal_Failure	Presence of chronic renal failure		Dichotomous
Risk_Cogn_Complaints	Presence of other cognitive complaints		Dichotomous
Diabetes	Presence of diabetes		Dichotomous
Frailty	Frailty according to (Robinson, et al. 2009)		Dichotomous

Table 8: Variables for the definition of frailty.

```

/* Define frailty */
DATA poseData; SET poseData;
  Frailty_Points = 0;
  Frailty = 0;
  if Mini_Cog_Total ne . and Mini_Cog_Total <= 3 then Frailty_Points =
    Frailty_Points + 1;
  if Albumin ne . and Albumin <= 3.3 then Frailty_Points = Frailty_Points + 1;
  if History_Falls = '3' then Frailty_Points = Frailty_Points + 1;
  if Haematocrit ne . and Haematocrit < 0.35 then Frailty_Points = Frailty_Points
    + 1;
  if Functional_Status_BL ne '1' then Frailty_Points = Frailty_Points + 1;
  if sum(Risk_Ischemic_Heart_Disease, Risk_Cardiac_Arrhythmia,
    Risk_Chronic_Heart_Failure, Risk_Peri_Vascular_Disease, Risk_Hemiplegia,
    Risk_COPD, Risk_Chronic_Resp_Failure, Risk_Alcohol_Abuse, Risk_Cancer,
    Risk_Transpl_Organs, Risk_Dementia, Risk_Cerebrovascular_Disease,
    Risk_Mild_Cogn_Impair, Risk_Chron_Renal_Failure, Risk_Cogn_Complaints,
    Diabetes = 1) >= 3 then Frailty_Points = Frailty_Points + 1;
  if Frailty_Points >=4 then Frailty = 1;
RUN;

```

6.7.2 Multimorbidity

Following the definition by the (World Health Organization 2015), multimorbidity is defined as the presence of at least two chronic conditions. We consider the following comorbidities:

- Diabetes
- Ischemic heart disease
- Chronic renal failure
- Cardiac arrhythmia or heart blocks

- Chronic heart failure or cardiomyopathy
- Hemiplegia
- Chronic respiratory failure
- Cancer
- Dementia
- Mild cognitive impairment
- Peripheral vascular disease
- COPD
- Chronic alcohol abuse
- Transplanted organ(s)
- Cerebrovascular disease
- Other cognitive complaints

If there is no information on the presence of a comorbidity, it is interpreted as “not present”.

Variable	Description	Unit	Data type
Risk_Ischemic_Heart_Disease	Presence of Ischemic heart disease		Dichotomous
Risk_Cardiac_Arrhythmia	Presence of cardiac arrhythmia or heart blocks		Dichotomous
Risk_Chronic_Heart_Failure	Presence of chronic heart failure or cardiomyopathy		Dichotomous
Risk_Peri_Vascular_Disease	Presence of peripheral vascular disease		Dichotomous
Risk_Hemiplegia	Presence of hemiplegia		Dichotomous
Risk_COPD	Presence of COPD		Dichotomous
Risk_Chronic_Resp_Failure	Presence of chronic respiratory failure		Dichotomous
Risk_Alcohol_Abuse	Chronic alcohol abuse		Dichotomous
Risk_Cancer	Presence of cancer		Dichotomous
Risk_Transpl_Organs	Presence of transplanted organ(s)		Dichotomous
Risk_Dementia	Presence of dementia		Dichotomous
Risk_Cerebrovascular_Disease	Presence of cerebrovascular disease		Dichotomous
Risk_Mild_Cogn_Impair	Presence of mild cognitive impairment		Dichotomous
Risk_Chron_Renal_Failure	Presence of chronic renal failure		Dichotomous
Risk_Cogn_Complaints	Presence of other cognitive complaints		Dichotomous
Diabetes	Presence of diabetes		Dichotomous
Multimorbidity	The presence of at least two or more chronic conditions. Definition according to (World Health Organization 2015).		Dichotomous

Table 9: Variables for the definition of multimorbidity.

```

/* Define multimorbidity */
DATA poseData; SET poseData;
  Multimorbidity = 0;
  if sum(Risk_Ischemic_Heart_Disease, Risk_Cardiac_Arrhythmia,
    Risk_Chronic_Heart_Failure, Risk_Peri_Vascular_Disease, Risk_Hemiplegia,
    Risk_COPD, Risk_Chronic_Resp_Failure, Risk_Alcohol_Abuse, Risk_Cancer,
    Risk_Transpl_Organs, Risk_Dementia, Risk_Cerebrovascular_Disease,
    Risk_Mild_Cogn_Impair, Risk_Chron_Renal_Failure, Risk_Cogn_Complaints,
    Diabetes = 1) >= 2 then Multimorbidity = 1;
RUN;

```

6.7.3 Timed up and go test result

The mobility assessment from the timed up and go test is derived as in (Bischoff, et al. 2003). The test result is evaluated as normal mobility if the patient was able to perform the timed up and go test in 12 seconds or less. If the patient was not able to perform the timed up and go test or took more than 12 seconds to perform the test, the test result is evaluated as limited mobility.

Variable	Description	Unit	Data type
Timed_Test_Able	Was the patient able to perform the timed up and go test		Nominal
Timed_Test_Time	Result from the timed up and go test	seconds	Continuous
Timed_Up_Go	Evaluation of normal or limited mobility. Definition according to (Bischoff, et al. 2003).		Dichotomous

Table 10: Variables for the definition of mobility according to the timed up and go test result.

```

/* Define mobility according to timed up and go test result */
DATA poseData; SET poseData;

```

```

length Timed_Up_Go $17;
if Timed_Test_Able = 1 and Timed_Test_Time <= 12 then Timed_Up_Go = "normal
mobility";
if Timed_Test_Able = 1 and Timed_Test_Time > 12 then Timed_Up_Go = "limited
mobility";
if Timed_Test_Able = 2 then Timed_Up_Go = "";
if Timed_Test_Able = 3 then Timed_Up_Go = "limited mobility";
RUN;

```

6.8 MISSING DATA

There will be no missing data for the primary endpoint (as patients without any information on follow-up have a censored survival time at time 0).

Variables with a very high proportion of missing data (e.g. 70% missing values) will be excluded from all inferential analyses.

For the main analysis, all independent variables without a high proportion of missing data will be handled in the inferential analyses using multiple imputation. The number of imputations is set in dependence of the percentage of incomplete cases with at least 10 and at most 25 imputations. Missing values are imputed using the fully conditional specification (FCS) method as described by (Brand 1999) and (Van Buuren, et al. 2006) (procedure PROC MI with fcs method in SAS). The number of burn-in iterations is set to 20. Heavily skewed continuous variables are transformed if necessary. We also include the primary outcome to impute missing values of independent variables (see (Moons, et al. 2006) and (Liu and De 2015)). The plausibility of imputed values is checked by considering descriptive statistics of the imputed and observed values.

If, for some reason, a variable cannot be handled using multiple imputation the variable will either be excluded from the analysis or included despite the missing values if the proportion of missing values is small (e.g. smaller than 5%).

In a sensitivity analysis of the primary endpoint, no missing data will be imputed and all patients with missing covariates will be excluded from the Cox regression model.

```

/* Multiple imputation of missing independent variables */
PROC MI data = dataset_help nimpute = pctmissing(min=10 max=25) seed = 911023
out = dataset_mi_help;
class Sex Severity_Intervention Urgency_Intervention Frailty
Interventional_Category_9 Referring_Facility Transfusion_Plasma
Transfusion_Platelets Transfusion_Red_Blood_Cells Anaesthesia_Technique
Multimorbidity Premedication_Intervention Timed_Up_Go;
* If applicable: impute continuous variables;
fcs nbiter=20 reg(Age /* example */);
* If applicable: transform heavily skewed data;
transform log(Age /* example */);
* If applicable: imputed dichotomous and nominal variables;
fcs nbiter=20 discrim(Sex Frailty /* example */// classeffects = include);
* If applicable: impute ordinal variables;
fcs nbiter=20 logistic(Severity_Intervention /* example */// order =
internal);
* Missing variables and other variables to estimate missing values. The
order of variables states in which order missing variables are imputed.;
var Surv_Time_30 Surv_Status_30 Age Sex Severity_Intervention
Urgency_Intervention Frailty /* example */;
RUN;

```

7. DESCRIPTIVE STATISTICS

7.1 BASELINE CHARACTERISTICS

Descriptive statistics of all continuous baseline characteristics include the number of available observations (N), mean and standard deviation (SD) or median, as well as lower (Q1) and upper quartile (Q3) in the case of heavily skewed data.

Descriptive statistics of all dichotomous or nominal baseline characteristics include the number of available observations (N), frequency (n) and percentage (%).

Descriptive statistics of all ordinal baseline characteristics include the number of available observations (N), frequency (n) and percentage (%). If the ordinal characteristic is associated with established numerical values, mean values and standard deviations (SD) are given or median, as well as lower (Q1) and upper quartile (Q3).

A possible tabular summary of baseline characteristics is shown in Table 11.

Characteristic	N	Statistical measure
Age	N _{age}	Mean (±SD) or Median (Q1-Q3)
Sex (female)	N _{sex}	n (%)
Referring facility	N _{fac}	
Home/ Lives independent		n (%)
Other hospital		...
...		
ASA Score	N _{ASA}	Mean (±SD) or Median (Q1-Q3)
Class I		n (%)
Class II		...
Class III		...
...
...		

Table 11: Descriptive statistics of different types of baseline characteristics

7.2 PRIMARY ENDPOINT

Descriptive statistics of the primary endpoint include the overall number of events until day 30 and a Kaplan-Meier curve of the survival rate within the entire collective. The plot of the Kaplan-Meier curve includes information on the number of patients at risk and the cumulative number of events. If considered useful, a 95%-confidence interval will be added to the plot (using the log-log method). If applicable, the median survival time is estimated using the Kaplan-Meier method. The estimated survival rate for 30 days will be reported.

In addition, the all-cause mortality until day 30 is shown in dependence of the defined age categories.

```
#####
## All-cause mortality in entire cohort      ##
#####
km <- survival::survfit(survival::Surv(Surv_Time_30, Surv_Status_30) ~ 1,
  data = poseData, conf.type = "log-log")
## Estimated survival rate after 30 days
summary(km, times = c(0, 30))
## Kaplan-Meier curve
pp <- survminer::ggsurvplot(fit = km,
  xlab = "Time in days",
  ylab = "Overall survival",
```

```

        legend = "none",           # No legend
        legend.labs = "",         # Add legend labs
        conf.int = TRUE,         # Add confidence interval
        risk.table = TRUE,       # Add number at risk
        cumevents = TRUE)       # Add cum. no. of events

#####
## All-cause mortality by age category      ##
#####
km <- survival::survfit(survival::Surv(Surv_Time_30, Surv_Status_30) ~
  Age_Category, data = poseData)
## Estimated survival rate after 30 days
summary(km, times = c(0, 30))
## Kaplan-Meier curve
pp<- survminer::ggsurvplot(fit = km,
  xlab = "Time in days",
  ylab = "Overall survival",
  legend.title = "Age group", # Add legend title
  legend.labs = c("[80, 85)", "[85, 90)", "> 90)",
  # Add legend labs
  conf.int = TRUE,           # Add confidence interval
  risk.table = TRUE,         # Add number at risk
  cumevents = TRUE)         # Add cum. no. of events

```

7.3 SECONDARY ENDPOINTS

7.3.1 Continuous secondary endpoints

Descriptive statistics of all continuous endpoints include the number of available observations (N), mean and standard deviation (SD) or median, as well as lower (Q1) and upper quartile (Q3) in the case of heavily skewed data. The skewness of continuous endpoints is investigated using histograms and boxplots.

7.3.2 Dichotomous secondary endpoints

Descriptive statistics of all dichotomous endpoints include the number of available observations (N), frequencies (n) and percentages (%).

7.3.3 Nominal secondary endpoints

Descriptive statistics of all nominal endpoints include the number of available observations (N), frequencies (n) and percentages (%).

7.3.4 Ordinal secondary endpoints

Descriptive statistics of all ordinal endpoints include the number of available observations (N), frequencies (n) and percentages (%). If the ordinal endpoint is associated with established numerical values, mean values and standard deviations (SD) are given or median, as well as lower (Q1) and upper quartile (Q3).

7.4 SENSITIVITY ANALYSES

For the sensitivity analysis of the primary endpoint with maximum duration of follow-up available we use the same descriptive techniques as described in Section 7.2. Descriptive statistics of the primary endpoint include the overall number of events and a Kaplan-Meier curve for the entire cohort with information on the numbers at risk. If applicable, the median survival time is estimated using the Kaplan-Meier method.

8. INFERENCE STATISTICS

8.1 PRIMARY ENDPOINT

The primary endpoint is analysed using a multivariable Cox regression model with frailty term (PROC PHREG, PROC MI and PROC MIANALYZE). The multivariable Cox regression model includes the following fixed effects:

- age,
- sex,
- severity of intervention (3 categories),
- urgency of the intervention (3 categories),
- frailty as defined in section 6.7.1.

Other candidate variables are included only as independent variables in the multivariable Cox model if the p-value from the univariable Cox regression model is at most 0.25. In the case of missing data, the pooled (or median) p-value from the univariable Cox regression models based on the multiply imputed data sets is taken. For continuous and dichotomous candidate variables the pooled p-value using Rubin's rule is considered for model inclusion. For nominal and ordinal candidate variables with more than two categories, the median type 3 test p-value is considered for model inclusion. The following candidate variables are considered for potential inclusion as fixed effects in the multivariable Cox regression model.

- type of intervention as defined in section 6.6.2 (9 categories),
- referring facility (5 categories),
- transfusion of plasma during intervention,
- transfusion of platelets during intervention,
- transfusion of red blood cells during intervention,
- anaesthesia technique as defined in section 6.6.3 (4 categories),
- multimorbidity as defined in section 6.7.2,
- premedication before intervention (3 categories),
- mobility assessments from the timed up and go test as defined in section 6.7.3.

The univariable and multivariable Cox models include a random center effect to account for the correlation between patients from the same center (frailty model with lognormal frailty distribution).

The proportional hazards assumption of all independent (candidate) variables is examined graphically using Schoenfeld residuals. The Schoenfeld residuals are taken from univariable Cox models without frailty term (resulting from the multiply imputed data sets in the case of missing data). If the proportional hazards assumption of an independent variable seems greatly violated (in a majority of imputed data sets), an interaction effect between the independent variable and time is included in the multivariable Cox regression model with frailty term.

The p-values of all fixed effects from the multivariable Cox regression model are reported. Estimated hazard ratios and 95%-confidence intervals may also be reported if considered useful. For continuous and dichotomous variables pooled p-values and estimates from the multiply imputed datasets are derived using Rubin's rule (PROC MIANALYZE). For nominal and ordinal variables with more than 2 categories the median type 3 test p-value of all analyses is reported. If a categorical variable with more than two categories is found statistically significant according to this p-value, pairwise unadjusted comparisons are requested. These unadjusted comparisons are again pooled using Rubin's rule (PROC MIANALYZE) and the resulting pooled p-values are adjusted using the Hochberg method to counteract the problem of multiple comparisons (PROC MULTTEST).

```
/* Example: Examine proportional hazards assumption in simplified univariable
```

```

Cox models */
ods graphics on;
PROC PHREG data = poseData_mi zph(transform=identity);
  by _Imputation_;
  model Surv_Time_30*Surv_Status_30(0) = Age / ties=efron;
RUN;
ods graphics off;

```

```

/* Variable selection for multivariable Cox model with missing data
   Example 1: combine p-values for continuous or dichotomous variable */
PROC PHREG data = poseData_mi;
  by _Imputation_;
  class Center_ID / param = glm order = internal;
  model Surv_Time_30*Surv_Status_30(0) = Age / ties=efron;
  random Center_ID / dist = lognormal;
  ods output ParameterEstimates=_es_uni;
RUN;
PROC MIANALYZE data=_es_uni;
  by Parameter;
  modeleffects Estimate;
  stderr stderr;
RUN;

/* Variable selection for multivariable Cox model with missing data
   Example 2: display median p-value for ordinal or nominal variable with more
   than 2 categories */
PROC PHREG data = poseData_mi;
  by _Imputation_;
  class Center_ID Severity_Intervention / param = glm order = internal;
  model Surv_Time_30*Surv_Status_30(0) = Severity_Intervention / ties=efron;
  random Center_ID / dist = lognormal;
  ods output ModelANOVA=_t3_uni;
RUN;
PROC MEANS data=_t3_uni median;
  var ProbChiSq;
RUN;

```

```

/* Multivariable Cox regression model */
PROC PHREG data = poseData_mi;
  by _Imputation_;
  class Center_ID Sex Severity_Intervention Urgency_Intervention Frailty
  /* if p-value < 0.25 in univariable Cox model selection */
  Interventional_Category_9 Referring_Facility Transfusion_Plasma
  Transfusion_Platelets Transfusion_Red_Blood_Cells Anaesthesia_Technique
  Multimorbidity Premedication_Intervention Timed_Up_Go
  / param = glm order = internal;
  model Surv_Time_30*Surv_Status_30(0) =
  Age Sex Severity_Intervention Urgency_Intervention Frailty
  /* if p-value < 0.25 in univariable Cox model selection */
  Interventional_Category_9 Referring_Facility Transfusion_Plasma
  Transfusion_Platelets Transfusion_Red_Blood_Cells Anaesthesia_Technique
  Multimorbidity Premedication_Intervention Timed_Up_Go
  / ties = efron;
  random Center_ID / solution dist = lognormal;
  lsmeans Severity_Intervention Urgency_Intervention
  /* if p-value < 0.25 in univariable Cox model selection */
  Interventional_Category_9 Referring_Facility Anaesthesia_Technique
  Premedication_Intervention / pdiff cl;
  * Save output to obtain pooled estimates;
  ods output ParameterEstimates=_es ModelANOVA=_t3 diffs=_diff;
RUN;

/* Example 1: Combine estimates from continuous and dichotomous variables and
   pool p-values */
PROC SORT data=_es;
  by Parameter ClassVal0 Imputation ;

```

```

RUN;
PROC MIANALYZE data=_es;
  by Parameter ClassVal0;
  modeleffects Estimate;
  stderr stderr;
  ods output ParameterEstimates = es_mianal;
RUN;
* Exponentiate to obtain HRs and CIs;
DATA hr_mianal; SET _es_mianal;
  HR_comb = exp(Estimate);
  HR_LCL_comb = exp(LCLMean);
  HR_UCL_comb = exp(UCLMean);
  HR_pval_comb = Probt;
  keep Parameter ClassVal0 HR_comb HR_LCL_comb HR_UCL_comb HR_pval_comb;
RUN;
PROC PRINT data = hr_mianal;
RUN;

/* Example 2: Display median p-value for ordinal or nominal variables with more
  than 2 categories and combine estimates of pairwise differences */
* Display median p-value;
DATA _t3_sub; SET _t3;
  if Effect ne 'Urgency_Intervention' then delete; /* example */
RUN;
PROC MEANS data=_t3_sub median;
  var ProbChiSq ProbChiSq2;
RUN;
* Combine estimates of pairwise differences;
DATA _diff_sub; SET _diff;
  if Effect ne 'Urgency_Intervention' then delete; /* example */
  keep _Imputation StmtNo Effect Urgency_Intervention
    _Urgency_Intervention Estimate StdErr zValue Probz;
RUN;
PROC SORT data=_diff_sub;
  by Effect Urgency_Intervention _Urgency_Intervention _Imputation;
RUN;
PROC MIANALYZE data=_diff_sub;
  by Urgency_Intervention _Urgency_Intervention;
  modeleffects Estimate;
  stderr StdErr;
  ods output ParameterEstimates = es_mianal_sub;
RUN;
* Adjust p-value to account for multiple comparisons;
DATA _es_mianal_sub; SET _es_mianal_sub;
  rename Probt = RAW_P;
RUN;
PROC MULTTEST inpvalues=_es_mianal_sub hochberg out=_es_mianal_sub2 noprint;
RUN;
* Exponentiate to obtain HRs and CIs;
DATA hr_mianal; SET _es_mianal_sub2;
  HR_comb = exp(Estimate);
  HR_LCL_comb = exp(LCLMean);
  HR_UCL_comb = exp(UCLMean);
  drop Parm StdErr LCLMean UCLMean DF Min Max Theta0 tValue RAW_P;
  rename hoc_p = adj_pVal;
RUN;
PROC PRINT data = hr_mianal;
RUN;

```

9. PREPARATION AND QUALITY CONTROL OF PROGRAM CODE

9.1 STORAGE OF LOG FILES

All SAS and R log files will be stored at the Department of Anaesthesia and can be obtained upon request.

9.2 CONTROL OF PROGRAM

Comments in the SAS or R-code are used to explain the code.

10.CHANGES FROM STUDY PROTOCOL

No changes are defined.

11. REFERENCES

- Bischoff, H A, et al. "Identifying a cut-off point for normal mobility: a comparison of the timed 'up and go' test in community-dwelling and institutionalised elderly women." *Age and Ageing*, 2003: 315-320.
- Brand, J P L. "Development, implementation and evaluation of multiple imputation strategies for the statistical analysis of incomplete data sets." *Erasmus University Rotterdam. Dissertation*. 1999.
- Liu, Y, and A De. "Multiple Imputation by Fully Conditional Specification for Dealing with Missing Data in a Large Epidemiologic Study." *International Journal of Medical Research*, 2015.
- Moons, K G, R A Donders, Stijnen T, and Harrell F E. "Using the outcome for imputation of missing predictor values was preferred." *Journal of Clinical Epidemiology*, 2006: 1092 - 1101.
- Oresanya, L B, W L Lyons, and E Finlayson. "Preoperative assessment of the older patient: a narrative review." *Journal of the American Medical Association*, 2014: 2110-2120.
- Robinson, T N, et al. "Redefining geriatric preoperative assessment using frailty, disability and co-morbidity." *Annals of Surgery*, 2009.
- Van Buuren, S, J P L Brand, C G M Groothuis-Oudshoorn, and D B Rubin. "Fully conditional specification in multivariate imputation." *Journal of Statistical Computation and Simulation*, 2006.
- Wood, A M, I R White, and Royston P. "How should variable selection be performed with multiply imputed data?" *Statistics in Medicine*, 2008: 3227-3246.
- World Health Organization. *World report on ageing and health*. World Health Organization, 2015.